

JAERI - M
91-098

INTRODUCTION TO SAS ON VAX

June 1991

Otto KARDAUN*, Yukitoshi MIURA, Toshiaki MATSUDA
and Hiroshi TAMAI

JAERI-Mレポートは、日本原子力研究所が不定期に公刊している研究報告書です。
入手の間合わせは、日本原子力研究所技術情報部情報資料課（〒319-11茨城県那珂郡東海村）あて、お申しこしください。なお、このほかに財団法人原子力弘済会資料センター（〒319-11茨城県那珂郡東海村日本原子力研究所内）で複写による実費頒布をおこなっております。

JAERI-M reports are issued irregularly.

Inquiries about availability of the reports should be addressed to Information Division, Department of Technical Information, Japan Atomic Energy Research Institute, Tokai-mura, Naka-gun, Ibaraki-ken 319-11, Japan.

© Japan Atomic Energy Research Institute, 1991

編集兼発行 日本原子力研究所
印 刷 (株)原子力資料サービス

Introduction to SAS on VAX

Otto KARDAUN*, Yukitoshi MIURA, Toshiaki MATSUDA
and Hiroshi TAMAI

Department of Fusion Plasma Research
Naka Fusion Research Establishment
Japan Atomic Energy Research Institute
Naka-machi, Naka-gun, Ibaraki-ken

(Received May 21, 1991)

To analyse, among others, the H-mode data base, a new version (6.06) of the SAS system has been installed on the VAX 3200 Workstation at JFT-2M. In this report, we summarize how to use SAS interactively (i.e. in 'display manager mode') on this machine. By a didactical example program and its annotated output we illustrate some of the capabilities of SAS. The report is intended to facilitate the access to the SAS documentation by physicists interested in plasma physical applications.

Keywords: H-mode Database, SAS Version 6.06, VAX Workstation at JFT-2M,
Regression Analysis, Principal Component Analysis, Confinement
Time Scaling

* Max-Planck-Institut für Plasmaphysik

VAX ワークステーション上の SAS について

日本原子力研究所那珂研究所炉心プラズマ研究部
Otto KARDAUM*・三浦 幸俊・松田 俊明・玉井 広史

(1991 年 5 月 21 日受理)

H-モードデータベース等のデータ解析のために、新バージョンの SAS (ver. 6.06 統計解析用パッケージ) を VAX 3200 ワークステーションに導入した。ディスプレイマネージャーを用いた VAX 上での SAS の取り扱い、及びその有用性について、サンプルプログラムを作り JFT2M のデータを解析することでまとめた。

Contents

1. Introduction	1
2. How to Use SAS on VAX in Interactive Mode	1
3. The Example Program of SAS and Its Output	3
Acknowledgements	36
Useful SAS Documentation	36
References	36

目 次

1. はじめに	1
2. インタラクティブモードでの SAS の使い方	1
3. サンプルプログラムとその出力	3
謝 辞	36
有用な SAS のマニュアル	36
参考文献	36

1. Introduction

Recently, the scaling of the energy confinement time of H-mode has attracted much attention. The knowledge of how the energy confinement time depends on various parameters, is not only essential as an intermediate step in understanding the physics associated with the H-mode, but is also inevitable in predicting the performance of future machines. Hence, a cooperation was established, initiated by the ITER Team, between the groups of ASDEX, DIII-D, JET, JFT-2M, PBX-M, and PDX to construct a joint global confinement 'ITER' H-mode Database[1-4]. This dataset, containing about thousand discharges and 75 global plasma parameters, is presently oriented towards H-mode realized by the NBI heating in the divertor configuration.

There are several statistical packages to analyse such a type of dataset, BMDP, GLIM, SAS, SPSS, and, on microcomputers, SYSSTAT and various spreadsheet programs. From these, SAS is one of the most extensive and integrated tools among those that are available on a range of hardware configurations, and is tending to become a standard in several plasma physical laboratories.

In this report, we summarize how to use SAS on VAX 3200 at JFT-2M, which is, except for the setting of some laser printer characteristics for graphical output and of a very few device-specific resource optimization options, fully compatible with SAS version 6.06 on another machine. In section 2, the start-up procedure to use SAS interactively (i.e., in 'display manager mode') is described briefly. A physically motivated SAS example program with explanations and its output are shown in section 3.

2. How to use SAS on VAX in interactive mode.

In this section, the start-up procedure to use SAS interactively (i.e., in 'display manager mode') on the VAX 3200 at JFT-2M is shown briefly. Bold-face type is used to mark commands that are to be typed in (if no menu option or function key is used).

VAX specifics:

2.1. Create a new ReGIS VT28X window.

At first, the new window of REGIS VT28X should be created by clicking the left button of the mouse. This window permits graphical output on the screen. The size of window should be enlarged to use it effectively. This is done by clicking the corresponding option 'change size' in the menu area on the left top corner of the window.

2.2. Login on VAX.

At present, only two users can use SAS on the VAX 3200. The enabled accounts are: MATSUDA and KARDAUN.

1. Introduction

Recently, the scaling of the energy confinement time of H-mode has attracted much attention. The knowledge of how the energy confinement time depends on various parameters, is not only essential as an intermediate step in understanding the physics associated with the H-mode, but is also inevitable in predicting the performance of future machines. Hence, a cooperation was established, initiated by the ITER Team, between the groups of ASDEX, DIII-D, JET, JFT-2M, PBX-M, and PDX to construct a joint global confinement 'ITER' H-mode Database[1-4]. This dataset, containing about thousand discharges and 75 global plasma parameters, is presently oriented towards H-mode realized by the NBI heating in the divertor configuration.

There are several statistical packages to analyse such a type of dataset, BMDP, GLIM, SAS, SPSS, and, on microcomputers, SYSSTAT and various spreadsheet programs. From these, SAS is one of the most extensive and integrated tools among those that are available on a range of hardware configurations, and is tending to become a standard in several plasma physical laboratories.

In this report, we summarize how to use SAS on VAX 3200 at JFT-2M, which is, except for the setting of some laser printer characteristics for graphical output and of a very few device-specific resource optimization options, fully compatible with SAS version 6.06 on another machine. In section 2, the start-up procedure to use SAS interactively (i.e., in 'display manager mode') is described briefly. A physically motivated SAS example program with explanations and its output are shown in section 3.

2. How to use SAS on VAX in interactive mode.

In this section, the start-up procedure to use SAS interactively (i.e., in 'display manager mode') on the VAX 3200 at JFT-2M is shown briefly. Bold-face type is used to mark commands that are to be typed in (if no menu option or function key is used).

VAX specifics:

2.1. Create a new ReGIS VT28X window.

At first, the new window of REGIS VT28X should be created by clicking the left button of the mouse. This window permits graphical output on the screen. The size of window should be enlarged to use it effectively. This is done by clicking the corresponding option 'change size' in the menu area on the left top corner of the window.

2.2. Login on VAX.

At present, only two users can use SAS on the VAX 3200. The enabled accounts are: MATSUDA and KARDAUN.

2.3. Input @sasini.

The 'sasini.com' is a file containing a few commands to initialize the SAS environment for the present system.. The file is activated by typing @sasini .
(The commands are actually carried out when SAS is started.)

General SAS statements:

2.4. Input sas.

This is the command to start SAS and go into its 'display manager'. After this command, you see the "LOG" and the "PROGRAM EDITOR" window of the display manager on the screen.

2.5. Input pmenu.

There are essentially 3 ways to input a command to SAS when using the display manager, (1) by typing a command on the command line, (2) by using the menu system activated by pmenu, (3) by hitting an appropriate function key. The menu system is a useful SAS feature . You can use the PF1+ ENTER key to switch between this PMENU SYSTEM and COMMAND INPUT. In addition to the "LOG", and "PROGRAM EDITOR" windows, some extra windows, such as "OUTPUT" and "KEYS", may be created during a session . The configuration (relative display) of these windows can be changed by pushing *Cascade, Tile* and so on in the *View* menu. There are important keys to show each window to the top.

F11	go to previous window
F12	go to next window
F17	go to the "LOG" window
F18	go to the "PROGRAM EDITOR" window
F19	go to the "OUTPUT" window

A list of all available key settings are shown in the "KEYS" window, which can be activated by typing *Keys* in the command line (or by choosing *Keys* from the *Help* menu, or by pressing PF20). For more information, see the SAS Language manual, Chapter 7 and/or the SAS on-line *Help* menu.

2.6. The next step is to edit and run a SAS program. To open a file on the "PROGRAM EDITOR" window, choose *Open* in the *File* menu. An overview of the SAS Editor is given in SAS Language manual, Chapter 8. Important keys for editing are the following ones (from the numerical key pad):

4 key	page up
5 key	page down
7 key	go to the top
8 key	go to the bottom.

From the "PROGRAM EDITOR" window, one can submit a VMS command as follows


```
00001 x ..... ;
```

where is a VMS command. Do not forget to type the ';', which marks the end of a SAS statement. (As usual in an IBM environment, the line-numbering region of the editor serves to place line editing commands, like **c** for: copy, **a** for: after this line, **m** for: move, **ts** for: split, upon pushing return, the line at the cursor position, etc.)

- 2.7. To print out, to save a file and to clear text on a window, choose the item *Print* in the *File* menu, *Save as ...* in the *File* menu and *Clear text* in the *Edit* menu, respectively. A VMS filename must be typed with quotes, for instance 'myfirst_file.sas'. (SAS strips off the quotes before passing the filename to VMS.)
- 2.8. After making a program, you can submit it by choosing *Submit* from the *Locals* menu, or by pressing the PF3 key. After that, the program will run. You can see the step that is being carried out in the right top corner on the window.
- 2.9. A listing of all submitted statements and of any errors (either syntax or run-time errors) that occurred can be viewed in the LOG window, which, as you know, can directly be accessed by pushing F17 key.
- 2.10. After running a program, the "OUTPUT" window is automatically pushed to the top. There is a useful tool to scan through various parts of the output. This is *Output Manager* called from the *Globals* menu. After that, the F19 key is the switch between OUTPUT and OUTPUT MANAGER.

3. The Example program of SAS and its output

The usage of SAS is shown by the following Example program and its output. The program should be largely self-explaining by the interspersed comments. (All text between * and the next semicolon, and all text between /* and */ is considered as comment, and hence skipped, by the SAS language interpreter.) The comments that have been added afterwards (for didactical purposes only) to the output file are typed in italic font.

3.1. The input text of the Example program.

```
*****;
*PROGRAM   : SASEXAM.SAS                               *;
*AUTHOR(S) : O. KARDAUN, YU. MIURA (JFT2M / IPP)      *;
*DATE      : 5 MAY 1991                                *;
*UPDATE    :                                           *;
```

```
00001 x ..... ;
```

where is a VMS command. Do not forget to type the ';', which marks the end of a SAS statement. (As usual in an IBM environment, the line-numbering region of the editor serves to place line editing commands, like **c** for: copy, **a** for: after this line, **m** for: move, **ts** for: split, upon pushing return, the line at the cursor position, etc.)

- 2.7. To print out, to save a file and to clear text on a window, choose the item *Print* in the *File* menu, *Save as ...* in the *File* menu and *Clear text* in the *Edit* menu, respectively. A VMS filename must be typed with quotes, for instance 'myfirst_file.sas'. (SAS strips off the quotes before passing the filename to VMS.)
- 2.8. After making a program, you can submit it by choosing *Submit* from the *Locals* menu, or by pressing the PF3 key. After that, the program will run. You can see the step that is being carried out in the right top corner on the window.
- 2.9. A listing of all submitted statements and of any errors (either syntax or run-time errors) that occurred can be viewed in the LOG window, which, as you know, can directly be accessed by pushing F17 key.
- 2.10. After running a program, the "OUTPUT" window is automatically pushed to the top. There is a useful tool to scan through various parts of the output. This is *Output Manager* called from the *Globals* menu. After that, the F19 key is the switch between OUTPUT and OUTPUT MANAGER.

3. The Example program of SAS and its output

The usage of SAS is shown by the following Example program and its output. The program should be largely self-explaining by the interspersed comments. (All text between * and the next semicolon, and all text between /* and */ is considered as comment, and hence skipped, by the SAS language interpreter.) The comments that have been added afterwards (for didactical purposes only) to the output file are typed in italic font.

3.1. The input text of the Example program.

```
*****;
*PROGRAM   : SASEXAM.SAS                      *;
*AUTHOR(S) : O. KARDAUN, YU. MIURA (JFT2M / IPP) *;
*DATE      : 5 MAY 1991                       *;
*UPDATE    :                                  *;
```

```

*INPUT FILES: BDAT.ITHMT1
*OUTPUT FILES:
/* DIDACTICAL PROGRAM TO ILLUSTRATE THE USE OF SAS,
   USING JFT2M DATA FROM THE ITER H-MODE DATABASE. */
/* (NOTATION: WE USE UPPER-CASE AS STANDARD FONT FOR THE
   COMMENTS, AND LOWER-CASE AS STANDARD FONT FOR THE
   ACTUAL SAS STATEMENTS. THE COMMENTS USUALLY PRECEDE THE
   SAS STATEMENTS.)
   */

*SETTING THE MAXIMAL BUFSIZE, AS BELOW, SAVES EXECUTION TIME IF
YOU HAVE TO READ LARGE FILES.;
*WHILE NUMBERING THE OUTPUT PAGES, START on hard copy WITH PAGE
NUMBER 1;;
options bufsize=65024 pageno=1;

*TITLES ARE PRINTED AS COMMENTARY LINES ONTO EACH OUTPUT PAGE
(UNTIL YOU CHANGE THE TITLE);
title 'program sasexam.sas';

*TO PRINT AN OVERVIEW OF THE VARIABLES AVAILABLE IN THE SAS
DATASET 'ITHMT1' (ACTUALLY CALLED: ITHMT1.SASEB$DATA;*) IN
YOUR VMS DIRECTORY;;
*(THE NAME SASUSER POINTS TO YOU OWN DIRECTORY);
title2 'procedure: proc contents data=sasuser.ithmt1;';
proc contents data=sasuser.ithmt1;

*DEFINITION TO DEFINE A NUMERICAL CONSTANT;
%let pi=3.14159;

*CREATE A NEW TEMPORARY DATASET (WITH THE NAME JFT2M)::
data JFT2M;
  *THE DATA IN THE FILE JFT2M ARE STORED USING
  4 BYTES PER (REAL OR INTEGER) NUMBER. THE DEFAULT IS
  8 BYTES;;
  length default=4;
  *USE SASUSER.ITHMT1 AS INPUT DATASET;;
  set sasuser.ithmt1;
  *USE ONLY THE 'STANDARD' ITER SELECTION (WHICH WAS STORED
  INTO THE BINARY VARIABLE PHYssel)::
  *IN ADDITION TO THIS RESTRICTION, ONLY THE JFT2M
  OBSERVATIONS ARE SELECTED;;
  if physsel=1;
  if tok='JFT2M';
  *KEEP ONLY THE VARIABLES THAT ARE MENTIONED IN THE KEEP
  STATEMENT (THE DEFAULT IS TO KEEP ALL VARIABLES)::

```

```

keep
  tok  shot  time phase
  bt ip    nel nev pinj pabs
  rgeo amin kappa
  taudia  taumhd wdia  wmhd
  gas dn  elm
  ;
run;

*CALCULATE THE MEAN VALUES AND SOME OTHER SUMMARY STATISTICS
  FOR THE SPECIFIED VARIABLES;;
title2 'procedure: proc means data=jft2m;';
proc means data=jft2m;
  var
    taudia  taumhd  wdia  wmhd
    bt ip    nel nev  pinj  pabs
    rgeo amin kappa
    gas dn  elm
  ;
run;

*MAKE AN EXTENDED DATASET (CALLED JFT2ME) WITH SOME EXTRA,
  DERIVED VARIABLES;
data jft2me;
  length default=4;
  set jft2m;
  ltaudia=log(taudia);
  ltaumhd=log(taumhd);
  ltmmd=ltaumhd-ltaudia;
  *USING 'ENGINEERING UNITS';;
  elbt=log(bt);
  elip=log(ip/1.0e6);
  elnel=log(nel/1.0e20);
  elnev=log(nev/1.0e20);
  elpinj=log(pinj/1.0e6);
  elpabs=log(pabs/1.0e6);
  elwdia=log(wdia/1.0e6);
  elwmhd=log(wmhd/1.0e6);
  *TO USE THE CONSTANT PI, DEFINED ABOVE;;
  elbtr=elbt-log(2*pi*rgeo);
  *MAKE A DESCRIPTIVE LABEL FOR VARIABLES FOR WHICH IT IS
    THOUGHT USEFUL;.
  *(VARIABLE NAMES CANNOT BE LONGER THAN 8 CHARACTERS,
    LABEL NAMES MAY BE UP TO 40 CHARACTERS. LONG LABELS MAY
    SOMETIMES PRODUCE AN UGLY OUTPUT APPEARANCE, HOWEVER.);

```

```

label ltmmd = 'log(taumhd/taudia)'
      elm= 'elms (0=no,1=yes)'
      gas='1=H, 2=D (av. of plasma and beam)'
      dn= 'config (0=SN,1=DN)';

run;

*PRESENT SOME MORE DETAILED UNIVARIATE STATISTICS,
  INCLUDING SOME LINEPRINTER GRAPHS,
  OF THE SPECIFIED VARIABLES;;
title2 'procedure: proc univariate data=jft2me plot;';
proc univariate data=jft2me plot;
  var  taudia  taumhd  ltmmd;
run;

*MAKE A FREQUENCY TABLE: ELM VERSUS
  ALL COMBINATIONS OF DN AND GAS, INCLUDING TOTALS;;
*(FOR FURTHER INFORMATION ABOUT SUCH, AND MORE
  COMPLICATED, TABLES SEE THE SAS PROCEDURES
  GUIDE, CHAPTER 37);
title2 'procedure: proc tabulate data = jft2me, etc.';
proc tabulate data = jft2me format=6. noseps;
  class dn gas elm;
  table (elm all), (dn*gas all);
run;

*MAKE THE CORRELATION MATRIX FOR SOME VARIABLES;;
*(THAT ARE THOUGHT TO BE OF POSSIBLE USE AS
  CONFINEMENT TIME REGRESSORS);
title2 'procedure: proc corr data=jft2me;';
proc corr data=jft2me;
  var elbt elip elnel elpabs;
run;

*PERFORM A PRINCIPAL COMPONENT ANALYSIS FOR THESE
  VARIABLES;;
*(BASED ON THE CORRELATION MATRIX, WHICH MEANS THAT
  ONE STANDARD DEVIATION, AS ESTIMATED FROM THE DATASET,
  IS CONSIDERED EQUALLY IMPORTANT FOR ALL VARIABLES);
title2 'procedure: proc princomp data=jft2me;';
proc princomp data=jft2me;
  var elbt elip elnel elpabs;
run;

*PERFORM A PRINCIPAL COMPONENT ANALYSIS FOR THESE
  VARIABLES;;
*(NOW BASED ON THE COVARIANCE MATRIX, WHICH MEANS THAT
  A CONSTANT FACTOR, FOR INSTANCE E=2.718... ,
  IS CONSIDERED EQUALLY IMPORTANT FOR ALL VARIABLES);

```

```

title2 'procedure: proc princomp data=jft2me covariance;';
proc princomp data=jft2me covariance;
  var elbt elip elnel elpabs;
run;

*LET US PERFORM A REGRESSION ANALYSIS;;
*(BECAUSE THE 'STANDARD SELECTION' CHOPPED OFF
  THE EXISTING B_T SCANS IN THE DATA,
  THE ABOVE 4 VARIABLES ARE ILL-CONDITIONED.
  FOR DIDACTICAL REASONS, WE NOW MAKE THE (STRONG) ASSUMPTION
  THAT THERE IS NO B-DEPENDENCE, AND OMIT B FROM THE
  REGRESSION. FURTHERMORE, WE RESTRICT THE EXPONENT OF THE
  ISOTOPE EFFECT TO BE 0.5.);
title2 'procedure: proc reg data=jft2me;';
proc reg data=jft2me;
  model ltaumhd ltaudia = elip elpabs elnel gas;
  restrict gas=0.5;
run;
quit; *to stop proc reg entirely;

*WE MAKE A DATASET WITH ONLY THE ELM-FREE DISCHARGES,;
data jft2mnel;
  set jft2me;
  if elm=0;
run;

*RUN, FOR DOCUMENTATION, PROC CONTENTS,;
title2 'procedure: proc contents data=jft2mnel;';
proc contents data=jft2mnel;
run;

*AND PERFORM THE REGRESSION AGAIN, NOW FOR TAU(MHD) AND
  ELM-FREE DISCHARGES ONLY,
  APPLYING NO RESTRICTION ON THE EXPONENT FOR GAS;;
title2 'procedure: proc reg data=jft2mnel;';
proc reg data=jft2mnel;
  model ltaumhd = elip elpabs elnel gas;
run;
quit; *to stop proc reg entirely;

*WE VERIFY THAT THE CONDITION IS REASONABLE BY USING
  PROC PRINCOMP ON THE ELM-FREE DATASET, AND INCLUDING
  GAS AS AN EXTRA VARIABLE;;
*(BASED ON THE COVARIANCE MATRIX, WHICH NOW IMPLIES THAT
  CHANGING THE VARIABLE GAS ONE UNIT IS CONSIDERED
  AS IMPORTANT AS CHANGING THE OTHER VARIABLES
  BY A FACTOR E=2.718);

```

```

title2 'procedure: proc princomp data=jft2mnel covariance;';
proc princomp data=jft2mnel covariance;
  var elip elnel elpabs gas;
run;

*WE USE PROC GLM TO INVESTIGATE INTERACTION BETWEEN THE
  POWER DEPENDENCE AND THE ISOTOPE EFFECT;;
*(SUCH AN INTERACTION WAS SUSPECTED BY JFT2M FROM
  OFFSET-LINEAR TYPE SCALING ANALYSES);
*(FOR MORE INFORMATION ON PROC GLM, SEE THE SAS/STAT USER'S
  GUIDE, CHAPTER 24);
title2 'procedure: proc glm data=jft2mnel;';
proc glm data=jft2mnel;
  class gas;
  model ltaumhd = gas elip elnel elpabs(gas) /solution;
  output out=jftnelpr p=ltaumhdp r=ltaumhdr;
  *(The last statement is optional. It serves to make
    an output dataset, called jftnelpr, which consists of the
    original dataset (jft2mnel), to which are added the variables
    ltaumhdp and ltaumhdr. These contain the p(predicted) values
    of ltaumhd and the r(residual), i.e. observed minus predicted,
    values of ltaumhd, respectively.);
run;
*TO END THE GLM PROCEDURE;;
*(OTHERWISE THIS IS ONLY DONE BY THE NEXT DATA STEP OR PROC
  STEP, AND INTERMEDIATE 'OPTION STATEMENTS', ETC. WILL BE
  OVERLOOKED. THE SAME APPLIES TO PROC REG AND PROC GPLOT);
quit;

*NOW WE ARE GOING TO MAKE A RESIDUAL PLOT;;
*FOR MORE INFORMATION, SEE THE SAS/GRAPH USER'S MANUAL;
*(THE ANNOTATE DATASET IS USED TO DRAW THE LINE 'Y=X');

data anno;
  *data value for x, data value for y;
  retain xsys '2' ysys '2';
  function= 'move'; x=-4.5; y=-4.5; output;
  function= 'draw'; x=-2.5; y=-2.5; output;
run;

*FOR MAKING THE PLOT ON THE VT286 SCREEN;;
goptions reset=all;
goptions display cback=white device=vt286
  hsize = 16.0cm vsize = 16.0cm;

title 'MHD confinement for ELM-free JFT2M discharges';
title2 'Power-law scaling with interaction between gas and
  Pabs';

```

```

title3 'Observed versus predicted from proc GLM';
  *(THE PLOT INFORMATION IS STORED IN THE GRAPHICAL CATALOG
    jftnelpr);
proc gplot data=jftnelpr gout=sasuser.jftnelpr;
  *THE AXIS STATEMENT IS OPTIONAL, BUT MAKES A NEATER GRAPH;;
  axis1 order = (-4.5 to -2.5 by 0.5)
        value = (h=1.5)   width=1
        offset=(0,0)     length=8 cm;
  plot ltaumhd*ltaumhdp/ fr vaxis=axis1 haxis=axis1
        annotate=anno;
run;

*TO REPLAY THE PLOT FROM THE GRAPHICAL CATALOG;;
proc greplay igout=sasuser.jftnelpr;
run;

*FOR A GOOD HARD-COPY ON THE TEKLN03 LASER PRINTER,
  THE GOPTIONS ARE CHANGED;;
*(THE GRAPHICAL INFORMATION IS NOW STORED IN A SO-CALLED
  GSF FILE, WHICH WE IDENTIFY BY THE NAME SASEXAM);
filename sasexam 'sasexam.gsf';
goptions device=tekln01 gsfname=sasexam
        gsfname=replace
        hsize = 17.0cm vsize = 17.0cm
        nodisplay noprompt nocharacters ;

*AND GPLOT IS RUN AGAIN;;
proc gplot data=jftnelpr;
  axis1 order = (-4.5 to -2.5 by 0.5)
        value = (h=1.5)   width=1
        offset=(0,0)     length=10 cm;
  plot ltaumhd*ltaumhdp/ fr vaxis=axis1 haxis=axis1
        annotate=anno;
run;
quit;

*WE ISSUE A VMS COMMAND TO SEND THE GSF FILE TO THE LASER
  PRINTER;;
  x print/passall sasexam.gsf;
*****endsas*****;

```

Figure 1 shows the graphical output from the laser printer.

3.2. The SAS output from the Example program.

The SAS output from the Example program was transferred to a word-processor on Macintosh and provided with some explanatory comments, typed in *italic font*.

17:41 Monday, May 6, 1991 1

program sasexam.sas
 procedure: proc contents data=sasuser.ithmtl;

CONTENTS PROCEDURE

Data Set Name: SASUSER.ITHMT1
 Member Type: DATA
 Engine: V606
 Created: 17:46 Wednesday, April 10, 1991
 Last Modified: 17:47 Wednesday, April 10, 1991
 Data Set Type:
 Label:
 Observations: 3466
 Variables: 120
 Indexes: 0
 Observation Length: 512
 Deleted Observations: 0
 Compressed: NO

-----Engine/Host Dependent Information-----

Data Set Page Size: 65024
 Number of Data Set Pages: 29
 First Data Page: 1
 Max Obs per Page: 124
 Obs in First Data Page: 95
 Filename: USR\$COMMON:[KARDAUN]ITHMT1.SASEB\$DATA
 Disk Blocks Allocated: 3684

-----Alphabetic List of Variables and Attributes-----

#	Variable	Type	Len	Pos
14	AMIN	Num	4	60
20	AREA	Num	4	84
31	BEILI2	Num	4	148
32	BEIMHD	Num	4	152
75	BEPDIA	Num	4	328
33	BEPMHD	Num	4	156
105	BETACR	Num	4	448
34	BETMHD	Num	4	160
10	BGASA	Num	4	44
11	BGASZ	Num	4	48
68	BMDMDIA	Num	4	296
103	BORO	Num	4	440
71	BSOURCE	Num	4	312
27	BT	Num	4	132
101	CARB	Num	4	432
70	COCTR	Num	4	308
22	CONFIG	Char	8	92
73	DALFDV	Num	4	320

Len=4 means that 4 bytes
 are used per observation
 for the specified variable

(reals and integers are
 treated equally)

17:41 Monday, May 6, 1991 2

Program sasexam.sas
 procedure: proc contents data=sasuser.ithmt1;

CONTENTS PROCEDURE

#	Variable	Type	Len	Pos
72	DALFMP	Num	4	316
3	DATE	Num	4	12
18	DELTA	Num	4	76
24	DIVMAT	Char	8	108
120	DN	Num	4	508
36	DNELDT	Num	4	168
48	DWDIA	Num	4	216
84	DWDIA7P	Num	4	364
80	DWDIA7PB	Num	4	348
74	DWDIAPAR	Num	4	324
76	DWHC	Num	4	332
49	DWMHD	Num	4	220
85	DWMHD7P	Num	4	368
81	DWMHD7PB	Num	4	352
79	DWMHDPAR	Num	4	344
114	ELM	Num	4	484
44	ENBI	Num	4	200
109	ENGSEL	Num	4	464
110	ENGSEL2	Num	4	468
26	EVAP	Char	8	124
91	GAS	Num	4	392
115	GELM	Num	4	488
97	HIGHRAD	Num	4	416
83	HLTIME	Num	4	360
92	HMODE	Num	4	396
19	INDENT	Num	4	80
28	IP	Num	4	136
17	KAPPA	Num	4	72
7	LHTIME	Num	4	32
25	LIMMAT	Char	8	116
78	MARKFR	Num	4	340
77	MARKOK	Num	4	336
38	NEO	Num	4	176
39	NEOTSC	Num	4	180
35	NEL	Num	4	164
37	NEV	Num	4	172
106	NOBETLIM	Num	4	452
104	NOBORO	Num	4	444

17:41 Monday, May 6, 1991 3

program sasexam.sas
 procedure: proc contents data=sasuser.ithmtl;

CONTENTS PROCEDURE

#	Variable	Type	Len	Pos
102	NOCARB	Num	4	436
93	NODIAMIS	Num	4	400
116	NOELM	Num	4	492
107	NOFAST	Num	4	456
117	NOGELM	Num	4	496
98	NOHIRAD	Num	4	420
108	NOLOWQ9	Num	4	460
118	NOLOWQC	Num	4	500
94	NOMISSN	Num	4	404
100	NOPEL	Num	4	428
82	OLTIME	Num	4	356
46	PABS	Num	4	208
99	PEL	Num	4	424
69	PELLET	Char	8	300
89	PERIND	Num	4	384
47	PFLOSS	Num	4	212
8	PGASA	Num	4	36
9	PGASZ	Num	4	40
6	PHASE	Char	8	24
112	PHYSSEL	Num	4	476
113	PHYSSEL2	Num	4	480
45	PINJ	Num	4	204
43	POHM	Num	4	196
42	PRAD	Num	4	192
30	Q95	Num	4	144
90	QCYL	Num	4	388
12	RGEO	Num	4	52
13	RMAG	Num	4	56
15	SEPLIM	Num	4	64
4	SHOT	Num	4	16
96	SSTATDIA	Num	4	412
95	SSTATMHD	Num	4	408
119	STAT	Num	4	504
64	TAUDIA	Num	4	280
65	TAUMHD	Num	4	284
66	TAUTH1	Num	4	288
67	TAUTH2	Num	4	292
51	TEO	Num	4	228

17:41 Monday, May 6, 1991 4

program sasexam.sas
 procedure: proc contents data=sasuser.ithmtl;

CONTENTS PROCEDURE

#	Variable	Type	Len	Pos
52	TEOTSC	Num	4	232
50	TEV	Num	4	224
54	TIO	Num	4	240
5	TIME	Num	4	20
53	TIV	Num	4	236
1	TOK	Char	8	0
111	TPI	Num	4	472
2	UPDATE	Num	4	8
21	VOL	Num	4	88
87	VS20M19	Num	4	376
29	VSURF	Num	4	140
86	VSURF20	Num	4	372
88	VSURF19C	Num	4	380
23	WALMAT	Char	8	100
55	WDIA	Num	4	244
58	WEKIN	Num	4	256
63	WFANI	Num	4	276
62	WFFORM	Num	4	272
61	WFPAR	Num	4	268
60	WPPER	Num	4	264
59	WIKIN	Num	4	260
57	WKIN	Num	4	252
56	WMHD	Num	4	248
16	XPLIM	Num	4	68
40	ZEFF	Num	4	184
41	ZEFFNEO	Num	4	188

17:41 Monday, May 6, 1991 5

program sasexam.sas
 procedure: proc means data=jft2m;

Variable	N	Mean	Std Dev	Minimum	Maximum
TAUDIA	231	0.0255898	0.0098259	0.0121500	0.0654800
TAUMHD	231	0.0309063	0.0116260	0.0153900	0.0753800
WDIA	231	23043.20	4979.20	11710.00	36970.00
WMHD	231	28664.46	5757.57	13400.00	44660.00
BT	231	1.2611515	0.0271613	1.0669999	1.3239999
IP	231	218299.13	36616.31	136600.00	261700.00
NEL	231	4.2405626E19	.	2.1189999E19	7.2839998E19
NEV	104	3.421471E19	.	1.8159999E19	5.1219997E19
PINJ	231	951751.52	420647.32	180000.00	1610000.00
PABS	231	906048.92	402054.40	176700.00	1587000.00
RGeo	231	1.3029480	0.0082734	1.2859999	1.3239999
AMIN	231	0.2699065	0.0064435	0.2485000	0.2866000
KAPPA	231	1.3965367	0.0341486	1.3000000	1.4529999
GAS	231	1.3398268	0.2338115	1.0000000	1.5000000
DN	231	0	0	0	0
ELM	231	0.0562771	0.2309564	0	1.0000000

(no value for
 Std. Dev of nel and nev,
 because calculating the
 sum of squares for these
 huge numbers in MNSA
 units caused overflow)

17:41 Monday, May 6, 1991 6

program sasexam.sas
 procedure: proc univariate data=jft2me plot;

Univariate Procedure

Variable=TAUDIA

Moments

N	231	Sum Wgts	231
Mean	0.02559	Sum	5.91124
Std Dev	0.009826	Variance	0.000097
Skewness	1.227281	Kurtosis	1.760936
USS	0.173473	CSS	0.022206
CV	38.3976	Std Mean	0.000646
T:Mean=0	39.58238	Prob> T	0.0001
Num ^= 0	231	Num > 0	231
M(Sign)	115.5	Prob> M	0.0001
Sgn Rank	13398	Prob> S	0.0001

Quantiles (Def=5)

100% Max	0.06548	99%	0.05692
75% Q3	0.03047	95%	0.04451
50% Med	0.02336	90%	0.0394
25% Q1	0.01774	10%	0.01531
0% Min	0.01215	5%	0.01404
		1%	0.0127
Range	0.05333		
Q3-Q1	0.01273		
Mode	0.01426		

Extremes


Lowest	Obs	Highest	Obs
0.01215 (94)	0.05009 (36)
0.01244 (93)	0.05673 (30)
0.0127 (95)	0.05692 (29)
0.0129 (92)	0.06397 (31)
0.01332 (138)	0.06548 (27)

17:41 Monday, May 6, 1991 7

program sasexam.sas
 procedure: proc univariate data=jft2me plot;

Univariate Procedure

Variable=TAUDIA



17:41 Monday, May 6, 1991 8

program sasexam.sas
 procedure: proc univariate data=jfft2me plot;

Univariate Procedure

Variable=TAUMHD

Moments

N	231	Sum Wgts	231
Mean	0.030906	Sum	7.13935
Std Dev	0.011626	Variance	0.000135
Skewness	1.017839	Kurtosis	0.89138
USS	0.251739	CSS	0.031088
CV	37.61703	Std Mean	0.000765
T:Mean=0	40.40373	Prob> T	0.0001
Num ^= 0	231	Num > 0	231
M(Sign)	115.5	Prob> M	0.0001
Sgn Rank	13398	Prob> S	0.0001

Quantiles (Def=5)

100% Max	0.07538	99%	0.06705
75% Q3	0.03733	95%	0.05225
50% Med	0.02813	90%	0.04772
25% Q1	0.02171	10%	0.01839
0% Min	0.01539	5%	0.01702
Range	0.05999	1%	0.01599
Q3-Q1	0.01562		
Mode	0.01918		

Extremes



Lowest	Obs	Highest	Obs
0.01539 (138)	0.06126 (28)
0.01546 (95)	0.06666 (27)
0.01599 (93)	0.06705 (29)
0.01603 (92)	0.06881 (30)
0.0161 (94)	0.07538 (31)

17:41 Monday, May 6, 1991 9

program sasexam.sas
 procedure: proc univariate data=jft2me plot;

Univariate Procedure

Variable=TAUMHD



17:41 Monday, May 6, 1991 10

Program sasexam.sas
 procedure: proc univariate data=jfft2me plot;

Univariate Procedure

Variable=LTMMD log(taumhd/taudia)

Moments

N	231	Sum Wgts	231
Mean	0.189338	Sum	43.73699
Std Dev	0.073461	Variance	0.005397
Skewness	0.186135	Kurtosis	0.049567
USS	9.522269	CSS	1.241212
CV	38.79914	Std Mean	0.004833
T:Mean=0	39.17273	Prob> T	0.0001
Num ^= 0	231	Num >	231
M(Sign)	115.5	Prob> M	0.0001
Sgn Rank	13398	Prob> S	0.0001

Quantiles (Def=5)

100% Max	0.394839	99%	0.384337
75% Q3	0.235441	95%	0.31701
50% Med	0.182599	90%	0.288963
25% Q1	0.143702	10%	0.104661
0% Min	0.002579	5%	0.065341
Range	0.39226	1%	0.01786
Q3-Q1	0.091739		
Mode	0.002579		

50% of all log(taumhd) values
 are between 0.144 and 0.235 higher
 than the corresponding log(taudia)
 values

Extremes

Lowest	Obs	Highest	Obs
0.002579 (48)	0.341568 (168)
0.015175 (50)	0.347805 (155)
0.01786 (27)	0.384337 (24)
0.032609 (51)	0.389721 (170)
0.03756 (219)	0.394839 (164)



17:41 Monday, May 6, 1991 11

program sasexam.sas
 procedure: proc univariate data=jft2me plot;

Univariate Procedure

Variable=ITMMD log(taumhd/taudia)

Stem Leaf	#
38 405	3
36	
34 28	2
32 59124	5
30 045899578	10
28 01222789036	11
26 02299345	8
24 124446913556699	15
22 00011122689112233355667	22
20 1223448902336777	16
18 11222333455568901233456777889	29
16 01122223344667888990123467789	29
14 1112446689013344556666678899	28
12 123334456677800001119	21
10 01570144679	11
8 36188	5
6 0259996	7
4 6725	4
2 38	2
0 358	3



Multiply Stem.Leaf by 10**2

17:41 Monday, May 6, 1991 12

program sasexam.sas
 procedure: proc tabulate data = jft2me, etc.

	config (0=SN,1=DN)	
	0	
	1=H, 2=D (av. of plasma and beam)	
	1 1.5 ALL	
	N N N	
elms (0=no,1=yes)	68 150 218	
0	6 7 13	
1	74 157 231	
ALL		

(There are no DN shots
 in the standard subset)

17:41 Monday, May 6, 1991 13

program sasexam.sas
 procedure: proc corr data=jft2me;

Correlation Analysis


4 'VAR' Variables: ELBT ELIP ELNEL ELPABS

Simple Statistics

Variable	N	Mean	Std Dev	Sum	Minimum	Maximum
ELBT	231	0.231781	0.022455	53.541453	0.064851	0.280657
ELIP	231	-1.537780	0.184644	-355.227089	-1.990698	-1.340556
ELNEL	231	-0.884500	0.233394	-204.319587	-1.551641	-0.316905
ELPABS	231	-0.221139	0.534223	-51.083027	-1.733302	0.461845

Pearson Correlation Coefficients / Prob > |R| under Ho: Rho=0 / N = 231

	ELBT	ELIP	ELNEL	ELPABS
ELBT	1.00000 0.0	0.20571 0.0017	0.02815 0.6704	-0.01738 0.7928
ELIP	0.20571 0.0017	1.00000 0.0	0.51350 0.0001	-0.28366 0.0001
ELNEL	0.02815 0.6704	0.51350 0.0001	1.00000 0.0	0.07916 0.2307
ELPABS	-0.01738 0.7928	-0.28366 0.0001	0.07916 0.2307	1.00000 0.0



17:41 Monday, May 6, 1991 14

Program sasexam.sas
 procedure: proc princomp data=jft2me;

Principal Component Analysis

231 Observations
 4 Variables

Simple Statistics

	ELBT	ELIP	ELNEL	ELPABS
Mean	0.2317811838	-1.537779607	-.8845003750	-.2211386459
Std	0.0224554357	0.184644119	0.2333940292	0.5342227406

Correlation Matrix

	ELBT	ELIP	ELNEL	ELPABS
ELBT	1.0000	0.2057	0.0282	-.0174
ELIP	0.2057	1.0000	0.5135	-.2837
ELNEL	0.0282	0.5135	1.0000	0.0792
ELPABS	-.0174	-.2837	0.0792	1.0000

Eigenvalues of the Correlation Matrix

	Eigenvalue	Difference	Proportion	Cumulative
PRIN1	1.60544	0.536161	0.401360	0.40136
PRIN2	1.06928	0.099490	0.267320	0.66868
PRIN3	0.96979	0.614294	0.242447	0.91113
PRIN4	0.35549		0.088874	1.00000

Eigenvectors

	PRIN1	PRIN2	PRIN3	PRIN4
ELBT	0.277620	-.152283	0.930274	-.185277
ELIP	0.714621	-.051243	-.083713	0.692591
ELNEL	0.584183	0.518152	-.206912	-.589437
ELPABS	-.266408	0.840061	0.291154	0.372228

17:41 Monday, May 6, 1991 15

program sasexam.sas
 procedure: proc princomp data=jft2me covariance;

Principal Component Analysis

231 Observations
 4 Variables

Simple Statistics

	ELBT	ELIP	ELNEL	ELPABS
Mean	0.2317811838	-1.537779607	-.8845003750	-.2211386459
Std	0.0224554357	0.184644119	0.2333940292	0.5342227406

Covariance Matrix

	ELBT	ELIP	ELNEL	ELPABS
ELBT	0.0005042466	0.0008529379	0.0001475522	-.0002084386
ELIP	0.0008529379	0.0340934505	0.0221292824	-.0279810097
ELNEL	0.0001475522	0.0221292824	0.0544727729	0.0098697681
ELPABS	-.0002084386	-.0279810097	0.0098697681	0.2853939366

Total Variance = 0.3744644065

Eigenvalues of the Covariance Matrix

	Eigenvalue	Difference	Proportion	Cumulative
PRIN1	0.288707	0.220253	0.770987	0.77099
PRIN2	0.068454	0.051627	0.182805	0.95379
PRIN3	0.016827	0.016352	0.044937	0.99873
PRIN4	0.000476	.	0.001270	1.00000

As the square root of the smallest eigenvalue is not small with respect to the suspected measurement error in (essentially) ELBT, see below in the table of eigenvectors, we have an ill-conditioned system of regression variables. (In fact, the restriction on q95 in the 'standard' sub-selection of the data has chopped off the existing Bt scans. for JFT-2M).


17:41 Monday, May 6, 1991 16

program sasexam.sas
 procedure: proc princomp data=jft2me covariance;

Principal Component Analysis

Eigenvalues

	PRIN1	PRIN2	PRIN3	PRIN4
ELBT	-.001018	0.008344	0.037987	0.999243
ELIP	-.106453	0.524582	0.843886	-.036570
ELNEL	0.031818	0.850827	-.524325	0.012861
ELPABS	0.993808	0.028960	0.107220	-.003306



17:41 Monday, May 6, 1991 17
 program sasexam.sas
 procedure: proc reg data=jft2me;

Model: MODEL1

NOTE: Restrictions have been applied to parameter estimates.

Dependent Variable: LTAUMHD

Analysis of Variance

Source	DF	Sum of Squares	Mean Square	F Value	Prob>F
Model	3	25.48872	8.49624	511.430	0.0001
Error	227	3.77109	0.01661		
C Total	230	29.25980			

Root MSE estimates 0.12889 R-square 0.8711
 one standard deviation Dep Mean -3.54171 Adj R-sq 0.8694
 of the residuals C.V. -3.63921
 (in our case on logarithmic scale)

Parameter Estimates

Variable	DF	Parameter Estimate	Standard Error	T for H0: Parameter=0	Prob > T	Variable Label
INTERCEP	1	-3.077716	0.07636343	-40.304	0.0001	Intercept
ELIP	1	0.931844	0.05796787	16.075	0.0001	
ELPABS	1	-0.375544	0.01724636	-21.775	0.0001	
ELNEL	1	-0.244227	0.04411456	-5.536	0.0001	
GAS	1	0.500000	0.00000000	.	.	1=H, 2=D (av. of plasma and beam)
RESTRICT	-1	-2.900369	0.39162759	-7.406	0.0001	

The intercept is to be interpreted as $\log(\tau(\text{mhd}))$ at engineering units (1 MA, 1 MW, $10^{*}20/\text{m}^{*}3$), and $\text{gas}=0$ (for $\text{gas}=1$, add 0.5 to the intercept). The coefficient needed for the restriction is more than 7 times as large as its standard error, which means that imposing the exponent 0.5 for gas is straining the fit of the model.

17:41 Monday, May 6, 1991 18
 Program sasexam.sas
 procedure: proc reg data=jft2me;

NOTE: Restrictions have been applied to parameter estimates.
 Dependent Variable: LTAUDIA

Analysis of Variance

Source	DF	Sum of Squares	Mean Square	F Value	Prob>F
Model	3	23.93862	7.97954	346.867	0.0001
Error	227	5.22205	0.02300		
C Total	230	29.16066			

A larger RMSE than for LTAUMHD
 Root MSE 0.15167 R-square 0.8209
 Dep Mean -3.73105 Adj R-sq 0.8186
 C.V. -4.06515

Parameter Estimates

Variable	DF	Parameter Estimate	Standard Error	T for H0: Parameter=0	Prob > T	Variable Label
INTERCEP	1	-3.469043	0.08986130	-38.604	0.0001	Intercept
ELIP	1	0.743379	0.06821417	10.898	0.0001	
ELPABS	1	-0.384828	0.02029480	-18.962	0.0001	
ELNEL	1	-0.142608	0.05191217	-2.747	0.0065	
GAS	1	0.500000	0.00000000	.	.	1=H, 2=D (av. of plasma and beam)
RESTRICT	-1	-2.246405	0.46085098	-4.874	0.0001	

The same comments as for LTAUMHD

17:41 Monday, May 6, 1991 19

program sasexam.sas
 procedure: proc contents data=jft2mnel;

CONTENTS PROCEDURE

```

Data Set Name: WORK.JFT2MNEL      Observations: 218
Member Type:  DATA              Variables: 32
Engine:      V606                 Indexes: 0
Created:    18:03 Monday, May 6, 1991  Observation Length: 136
Last Modified: 18:03 Monday, May 6, 1991 Deleted Observations: 0
Data Set Type:                               Compressed: NO
Label:
  
```

-----Engine/Host Dependent Information-----

```

Data Set Page Size: 65024
Number of Data Set Pages: 1
First Data Page: 1
Max Obs per Page: 439
Obs in First Data Page: 218
Filename: DUA0:[USRDSK.KARDAUN.SAS$WORK00000043]JFT2MNEL.SASEB$DATA
Disk Blocks Allocated: 128
  
```

-----Alphabetic List of Variables and Attributes-----

#	Variable	Type	Len	Pos	Label
6	AMIN	Num	4	28	
8	BT	Num	4	36	
20	DN	Num	4	84	config (0=SN,1=DN)
24	ELBT	Num	4	100	
32	ELBTR	Num	4	132	
25	ELIP	Num	4	104	
19	ELM	Num	4	80	elms (0=no,1=yes)
26	ELNEL	Num	4	108	
27	ELNEV	Num	4	112	
29	ELPABS	Num	4	120	
28	ELPINJ	Num	4	116	
30	ELWDIA	Num	4	124	
31	ELWMHD	Num	4	128	
18	GAS	Num	4	76	1=H, 2=D (av. of plasma and beam)
9	IP	Num	4	40	
7	KAPPA	Num	4	32	
21	LTAUDIA	Num	4	88	
22	LTAUMHD	Num	4	92	

17:41 Monday, May 6, 1991 20

program sasexam.sas
 procedure: proc contents data=jft2mnel;

CONTENTS PROCEDURE

#	Variable	Type	Len	Pos	Label
23	LTMMD	Num	4	96	log(taumhd/taudia)
10	NEL	Num	4	44	
11	NEV	Num	4	48	
13	PABS	Num	4	56	
4	PHASE	Char	8	16	
12	PINJ	Num	4	52	
5	RGEO	Num	4	24	
2	SHOT	Num	4	8	
16	TAUDIA	Num	4	68	
17	TAUMHD	Num	4	72	
3	TIME	Num	4	12	
1	TOK	Char	8	0	
14	WDIA	Num	4	60	
15	WMHD	Num	4	64	

17:41 Monday, May 6, 1991 21
 program sasexam.sas
 procedure: proc reg data=jft2mnel;

Model: MODEL1
 Dependent Variable: LTAUMHD

Analysis of Variance

Source	DF	Sum of Squares	Mean Square	F Value	Prob>F
Model	4	25.26414	6.31603	506.076	0.0001
Error	213	2.65833	0.01248		
C Total	217	27.92246			
Root MSE		0.11172	R-square	0.9048	
Dep Mean		-3.55581	Adj R-sq	0.9030	
C.V.		-3.14178			

Parameter Estimates

Variable	DF	Parameter Estimate	Standard Error	T for H0: Parameter=0	Prob > T	Variable Label
INTERCEP	1	-2.617078	0.08572272	-30.530	0.0001	Intercept
ELIP	1	0.915485	0.05270244	17.371	0.0001	
ELPABS	1	-0.458126	0.01742094	-26.297	0.0001	
ELNEL	1	-0.178928	0.04128467	-4.334	0.0001	
GAS	1	0.171682	0.03833003	4.479	0.0001	1=H, 2=D (av. of plasma and beam)

Indeed, the coefficient for GAS differs significantly from 0.5

17:41 Monday, May 6, 1991 22

Program sasexam.sas
 procedure: proc princomp data=jft2mnel covariance;

Principal Component Analysis

218 Observations
 4 Variables

	ELIP	ELNEL	ELPABS	GAS
Mean	-1.545772443	-0.8770125167	-0.1936721051	1.344036697
Std	0.185429192	0.2352187962	0.5299623770	0.232173141

Covariance Matrix

	ELIP	ELNEL	ELPABS	GAS
ELIP	0.0343839851	0.0244417087	-0.0256918823	0.0082595933
ELNEL	0.0244417087	0.0553278821	0.0045265850	0.0110093677
ELPABS	-0.0256918823	0.0045265850	0.2808601210	-0.0578147705
GAS	0.0082595933	0.0110093677	-0.0578147705	0.0533043673

1-H, 2-D (av. of plasma and beam)

Total Variance = 0.4244763555

Eigenvalues of the Covariance Matrix

	Eigenvalue	Difference	Proportion	Cumulative
PRIN1	0.297494	0.223206	0.700848	0.70085
PRIN2	0.074288	0.036941	0.175010	0.87586
PRIN3	0.037347	0.021998	0.087983	0.96384
PRIN4	0.015349	.	0.036159	1.00000

The condition is more or less OK. The data variation (1 std. dev.) is about 12.5% in the least favorable direction. Unfortunately, the uncertainty in isotope composition (which is correlated with that direction, see below) is also quite large for part of the data (i.e. if gas=1.5). Hence, the standard error of 0.04 may be somewhat optimistic, and a more accurate determination of the gas composition and/or a more elaborate statistical analysis which takes the uncertainty in the gas composition better into account, are expected to lead to better estimates for the isotope effect, than the one shown above. Nevertheless, it seems fair to state that the isotope effect in JFT-2M for ELM-free H-mode is expected to be somewhere between 0.1 and 0.3, under the assumption that there is no interaction between the effects of isotope composition and the other regression variables. Such an interaction was, however, suspected from previous the JFT-2M investigations ...

17:41 Monday, May 6, 1991 23

program sasexam.sas
 procedure: proc princomp data=jft2mnel covariance;

Principal Component Analysis

Eigenvectors

	PRIN1	PRIN2	PRIN3	PRIN4
ELIP	-.102012	0.481827	-.331411	0.804738
ELNEL	-.002817	0.818148	-.154521	-.553849
ELPABS	0.967079	0.122820	0.184499	0.125036
GAS	-.233118	0.288776	0.912277	0.173246

1=H, 2=D (av. of plasma and beam)

17:41 Monday, May 6, 1991 24

```
program sasexam.sas  
procedure: proc glm data=jft2mnel;
```

```
General Linear Models Procedure  
Class Level Information
```

Class	Levels	Values
GAS	2	1 1.5

```
Number of observations in data set = 218
```


17:41 Monday, May 6, 1991 25

```

program sasexam.sas
procedure: proc glm data=jft2mnel;

```

General Linear Models Procedure

Dependent Variable: LTAUMHD

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	5	25.32789045	5.06557809	413.90	0.0001
Error	212	2.59457153	0.01223854		
Corrected Total	217	27.92246198			
R-Square		C.V.	Root MSE		LTAUMHD Mean
	0.907079	-3.111185	0.11062796		-3.55581439

Source	DF	Type I SS	Mean Square	F Value	Pr > F
GAS	1	6.87733153	6.87733153	561.94	0.0001
ELIP	1	7.55642044	7.55642044	617.43	0.0001
ELNEL	1	2.19949193	2.19949193	179.72	0.0001
ELPABS (GAS)	2	8.69464656	4.34732328	355.22	0.0001
Source	DF	Type III SS	Mean Square	F Value	Pr > F
GAS	1	0.12929939	0.12929939	10.56	0.0013
ELIP	1	3.81026959	3.81026959	311.33	0.0001
ELNEL	1	0.22439749	0.22439749	18.34	0.0001
ELPABS (GAS)	2	8.69464656	4.34732328	355.22	0.0001

Parameter	Estimate	T for H0: Parameter=0	Pr > T	Std Error of Estimate
INTERCEPT	-2.341816984 B	-34.39	0.0001	0.06809745
GAS	-0.067172339 B	-3.25	0.0013	0.02066603
ELIP	0.000000000 B	.	.	.
ELNEL	0.922421381	17.64	0.0001	0.05227769
ELPABS (GAS) 1	-0.175198443	-4.28	0.0001	0.04091534
ELPABS (GAS) 2	-0.582390708	-10.20	0.0001	0.05711290

17:41 Monday, May 6, 1991 26

```

program sasexam.sas
procedure: proc glm data=jft2mnel;

```


General Linear Models Procedure

Dependent Variable: LTAUMHD

Parameter	Estimate	T for H0: Parameter=0	Pr > T	Std Error of Estimate
ELPABS (GAS) 1.5	-0.446961861	-24.93	0.0001	0.01793140

NOTE: The X'X matrix has been found to be singular and a generalized inverse was used to solve the normal equations. Estimates followed by the letter 'B' are biased, and are not unique estimators of the parameters.

Indeed, the exponents of Pabs in the scaling of tau(mhd) are significantly different between the two types of gases (gas=1.5 and gas=2.0). Note that we allowed for two different intercepts. (Their difference (-0.07) has to be interpreted at 1 engineering unit for (IP, Pabs, nel), i.e. (1 MA, 1 MW, 10^20/m^3), and gas=0. For other values of the plasma parameters, the intercept difference can easily be calculated by using the above coefficients. To calculate the standard deviation of this difference at those other values requires either the covariance matrix of the regression coefficients (not given here) or a new run of proc glm with rescaled variables. Note that it does not make much sense to ask in general for the statistical significance of the intercept difference. 'It depends on where you are'.



Acknowledgements

The first author (OK) would like to thank the JFT-2M group for their hospitality during his stay in Spring 1991. All authors are grateful to Dr. K. Thomsen for explaining to them, in an earlier stage, some of the finer aspects of the SAS GRAPH procedures at JET. They also wish to acknowledge Drs. N.Suzuki, H.Maeda, S.Shimamoto, S.Tamura and M.Yoshikawa for continuous support and encouragement.

Useful SAS[@] documentation

- SAS Institute Inc. (1990), *SAS Language and Procedures: Introduction*, Version 6, First Edition, Cary, NC.
- SAS Institute Inc. (1990), *SAS Language: Reference*, Version 6, First Edition, Cary, NC.
- SAS Institute Inc. (1990), *SAS Procedures Guide*, Version 6, Third Edition, Cary, NC.
- SAS Institute Inc. (1989), *SAS Language and Procedures: Usage*, Version 6, First Edition, Cary, NC.
- SAS Institute Inc. (1989), *SAS/STAT User's Guide: Vols. 1 and 2*, Version 6, Fourth Edition, Cary, NC
- SAS Institute Inc. (1990), *SAS/GRAPH Software:Introduction*, Version 6, First Edition, Cary, NC.
- SAS Institute Inc. (1990), *SAS/GRAPH User's Guide: Vols. 1 and 2*, Version 6, First Edition, Cary, NC.
- SAS Institute Inc. (1990), *SAS Companion for the VMS Environment* Version 6, First Edition, Cary, NC.

References

- [1] J.P. Christiansen et al., ITER-IL-PH-4-1-1 (1990).
- [2] Y. Miura et al., submitted to Plasma Physics and Controlled Fusion.
- [3] O. Kardaun et al., in: Controlled Fusion and Plasma Physics (Proc. 17th Eur. Conf.Amsterdam 1990) Vol 14B, Part I, EPS, 110.
- [4] P.J. Mc Carthy et al., IPP 5/34 (1990).

@ SAS is a registered trademark of SAS Institute Inc.

Acknowledgements

The first author (OK) would like to thank the JFT-2M group for their hospitality during his stay in Spring 1991. All authors are grateful to Dr. K. Thomsen for explaining to them, in an earlier stage, some of the finer aspects of the SAS GRAPH procedures at JET. They also wish to acknowledge Drs. N.Suzuki, H.Maeda, S.Shimamoto, S.Tamura and M.Yoshikawa for continuous support and encouragement.

Useful SAS[@] documentation

- SAS Institute Inc. (1990), *SAS Language and Procedures: Introduction*, Version 6, First Edition, Cary, NC.
- SAS Institute Inc. (1990), *SAS Language: Reference*, Version 6, First Edition, Cary, NC.
- SAS Institute Inc. (1990), *SAS Procedures Guide*, Version 6, Third Edition, Cary, NC.
- SAS Institute Inc. (1989), *SAS Language and Procedures: Usage*, Version 6, First Edition, Cary, NC.
- SAS Institute Inc. (1989), *SAS/STAT User's Guide: Vols. 1 and 2*, Version 6, Fourth Edition, Cary, NC
- SAS Institute Inc. (1990), *SAS/GRAPH Software:Introduction*, Version 6, First Edition, Cary, NC.
- SAS Institute Inc. (1990), *SAS/GRAPH User's Guide: Vols. 1 and 2*, Version 6, First Edition, Cary, NC.
- SAS Institute Inc. (1990), *SAS Companion for the VMS Environment* Version 6, First Edition, Cary, NC.

References

- [1] J.P. Christiansen et al., ITER-IL-PH-4-1-1 (1990).
- [2] Y. Miura et al., submitted to Plasma Physics and Controlled Fusion.
- [3] O. Kardaun et al., in: Controlled Fusion and Plasma Physics (Proc. 17th Eur. Conf.Amsterdam 1990) Vol 14B, Part I, EPS, 110.
- [4] P.J. Mc Carthy et al., IPP 5/34 (1990).

@ SAS is a registered trademark of SAS Institute Inc.

Acknowledgements

The first author (OK) would like to thank the JFT-2M group for their hospitality during his stay in Spring 1991. All authors are grateful to Dr. K. Thomsen for explaining to them, in an earlier stage, some of the finer aspects of the SAS GRAPH procedures at JET. They also wish to acknowledge Drs. N.Suzuki, H.Maeda, S.Shimamoto, S.Tamura and M.Yoshikawa for continuous support and encouragement.

Useful SAS[@] documentation

- SAS Institute Inc. (1990), *SAS Language and Procedures: Introduction*, Version 6, First Edition, Cary, NC.
- SAS Institute Inc. (1990), *SAS Language: Reference*, Version 6, First Edition, Cary, NC.
- SAS Institute Inc. (1990), *SAS Procedures Guide*, Version 6, Third Edition, Cary, NC.
- SAS Institute Inc. (1989), *SAS Language and Procedures: Usage*, Version 6, First Edition, Cary, NC.
- SAS Institute Inc. (1989), *SAS/STAT User's Guide: Vols. 1 and 2*, Version 6, Fourth Edition, Cary, NC
- SAS Institute Inc. (1990), *SAS/GRAPH Software:Introduction*, Version 6, First Edition, Cary, NC.
- SAS Institute Inc. (1990), *SAS/GRAPH User's Guide: Vols. 1 and 2*, Version 6, First Edition, Cary, NC.
- SAS Institute Inc. (1990), *SAS Companion for the VMS Environment* Version 6, First Edition, Cary, NC.

References

- [1] J.P. Christiansen et al., ITER-IL-PH-4-1-1 (1990).
- [2] Y. Miura et al., submitted to Plasma Physics and Controlled Fusion.
- [3] O. Kardaun et al., in: Controlled Fusion and Plasma Physics (Proc. 17th Eur. Conf.Amsterdam 1990) Vol 14B, Part I, EPS, 110.
- [4] P.J. Mc Carthy et al., IPP 5/34 (1990).

@ SAS is a registered trademark of SAS Institute Inc.

MHD confinement for ELM-free JFT2M discharges
 Power-law scaling with interaction between gas and Pabs
 Observed versus predicted from proc GLM




Fig.1 Graphical output from SAS.